The negotiation model of rhizobium-legume mutualism

05 July 2006
Erol Akcay and Joan Roughgarden

Department of Biological Sciences,
Stanford University
Stanford, CA

email: erol@stanford.edu, joan.roughgarden@stanford.edu

Part 1: Payoff Functions and the Nash Bargaining Solution

The payoff functions are derived from the rhizobium and legume models in the manuscript (ub and up correspond to rb and rp, respectively). There are two functions defined for each of them, one (ubparam, upparam) is used for symbolic calculation (of the pareto boundary) the other (ub, up) has the numerical values of parameters in it, and used in simulations.

```
ubparam[n_, c_] := Ic - Ia /. {Ia -> n, Ic -> c}; upparam[n_, c_] :=
  1/2 (Ia - Ic + k1 k2 + Lp - Sqrt[Ia^2 + 2 Ia (Ic + k1 k2 - Lp) + (-Ic + k1 k2 + Lp)^2]) /.
   {Ic -> c, Ia -> n, Lp -> l}; l = 1;
```

$$\text{upparam}[n\_, c\_] := \frac{1}{2}\left(Ia - Ic + k1\,k2 + Lp - \sqrt{Ia^2 + 2\,Ia\,(Ic + k1\,k2 - Lp) + (-Ic + k1\,k2 + Lp)^2}\right)$$

```
parameters = {k1 -> .1, k2 -> .1, k3 -> .1, k4 -> .1};

ub[n_, c_] := ubparam[n, c] /. parameters;
up[n_, c_] := upparam[n, c] /. parameters;

NashProduct[Ia_, Ic_] := ub[Ia, Ic] up[Ia, Ic];
```

The location of Nash Bargaining Solution for this set of parameters is given by (x and y are dummy variables for n and c, respectively):

```
nbs =
 Maximize[{NashProduct[Ia, Ic], Ia > 0, 1 > Ic > 0, Ic > Ia }, {Ia, Ic}]

{0.102612, {Ia → 0.261598, Ic → 0.738402}}
```

And the payoffs at the NBS:

```
{ub[Ia, Ic], up[Ia, Ic]} /. nbs[[2]]
```

```
{0.476805, 0.215207}
```

```
{ub[.5, .55], up[.5, .55]}
```

```
{0.05, 0.406515}
```

Part 2: The Simulation

```
history2 = Flatten[
   Reap[n = .5; c = 0.55;
     Do[
      Sow[{ub[n, c], up[n, c]}, payoff];
      Sow[{n, c}, strategy];

      (*Who gets to offer is determined at random*)
      rand = Random[Real, {0, 1}];
      If[rand ≤ .5,

       Sow[Null, srhizo]; Sow[Null, prob2];
               (*Plant is proposing,
        a random increase-or decrease in both fluxes is selected*)
       nprime = n + .05 Random[Real, {-1, 1}];
       cprime = c + .05 Random[Real, {-1, 1}];

       (*This checks that the payoffs are real-valued at the new
          values, if they're not it goes to the next step*)

       If[Im[up[nprime, cprime]] ≠ 0 ∨ Im[ub[nprime, cprime]] ≠ 0,
        Sow[0, prob]; Sow["invalid", outcome];
        Sow[.5, {splant, srhizo}]; Continue[], Null];

       upprime = up[nprime, cprime];
       ubprime = ub[nprime, cprime];

       (*If plant receives lower payoff from the new fluxes,
        it reverts to the previous fluxes,
        i.e. no change happens at this step*)

       If[upprime ≤ up[n, c],
        Sow[0.5, splant]; Sow[0, prob];
        Sow["bad offer plant", outcome]; Continue[]];

       (*If the new fluxes also give rhizobium an
         increase in payoff, they become accepted, otherwise,
        the waiting times p of the nodule and q of the plant
         is computed according to equation 11 in the manuscript*)

       If[ubprime > ub[n, c],
        n = nprime; c = cprime;
        Sow[0, prob]; Sow["mutual ben plant", outcome];
        Sow[1, splant]; Continue[]];
```

```
p = Min[1, (ub[n, c] - ubprime) / ub[n, c]];
q = Min[1, (upprime - up[n, c]) / upprime];
    Sow[Min[p, q], prob];

        (*Plant reverts to previous fluxes if the p selected
  by rhizobium is greater than (up'-up)/up' *)
    If[p ≥ q ,

     Sow[.5, splant]; Sow["conflict plant unsc", outcome] ,

     n = nprime; c = cprime;
 Sow["conflict plant sccsfl", outcome]; Sow[1, splant]],

        (*Rhizobium is proposing, a random increase
  of decrease is again drawn  for both fluxes*)
Sow[Null, splant]; Sow[Null, prob1];
nprime = n + .05 Random[Real, {-1, 1}];
cprime = c + .05 Random[Real, {-1, 1}];

(*This checks that the payoffs are real-valued at the new
   values, if they' re not it goes to the next step*)

If[Im[up[nprime, cprime]] ≠ 0 ∨ Im[ub[nprime, cprime]] ≠ 0 ,
 Sow[0, prob]; Sow["invalid", outcome];
 Sow[.5, {splant, srhizo}]; Continue[], Null];

upprime = up[nprime, cprime];
ubprime = ub[nprime, cprime];

        (*Like plant, rhizobium reverts
  immediately if these fluxes are better for it*)

If[
 ub[nprime, cprime] ≤ ub[n, c],
 Sow[.5, srhizo]; Sow[0, prob];
 Sow["bad offer rhizo", outcome]; Continue[]];

        (*If plant is also better off,
 it accepts right away, if it is not, plant and rhizobium
  select their waiting times, q and p, respectively*)

If[upprime > up[n, c],
 n = nprime; c = cprime;
 Sow[0, prob];
 Sow["mutual ben rhizo", outcome]; Sow[1, srhizo]; Continue[],

 q = Min[1, (up[n, c] - upprime) / up[n, c]];
```

```
          p = Min[1, (ubprime - ub[n, c]) / ubprime];
              Sow[Min[p, q], prob];

                  (*Rhizobium reverts
             to previous fluxes if p>(ub'-ub)/ub'*)

              If[q ≥ p,
               Sow[.5, srhizo]; Sow["conflict rhizo unsc", outcome],
               c = cprime; n = nprime;
           Sow["conflict rhizo sccsfl", outcome];
               Sow[1, srhizo]]]
        ]

        , {i, 1, 2000}],
       {payoff, strategy, prob, splant, srhizo, outcome}][[2]], 1];

(*End of simulation, the history is saved in history in the
  format: {{history of payoff pairs}, {history of flux pairs},
    {history of responding player' s threats},
    {whether plant was successful},
    {whether rhizobium was successful}, {history of the outcome as
      "immediate acceptance" or "conflict with rhizobium winning"}}*)
```